DTP2000 1

Introduction to ELAN and the rewrite calculus

Claude Kirchner LORIA and INRIA Nancy

Claude.Kirchner@loria.fr

Companion web site: www.loria.fr/ELAN Companion introduction to the ρ-calculus: www.loria.fr/~ckirchne/=rho Casting: Horatiu Cirstea, Hubert Dubois, Hélène Kirchner, Pierre-Etienne Moreau, Christophe Ringeissen

DTP2000 2

Rules are ubiquitous in Mathematics, Logic and Computer Science

Production rules

Program transformation rules Grammar rules Transition rules Logic programming rules Constraint manipulation rules Inference rules Type checking rules Derivative rules

and much more ...

DTP2000 3

Some examples

• the Unix mail system: look at the /etc/sendmail.cf file! • tree automaton • the so called natural semantics • Mathematica • Reve, Spass, Otter, Saturate • . . .

DTP2000 4

Some success

• re-engineering of Cobol programs [ASF+SDF, CWI Amsterdam] • proof of the Robbin’s conjecture [New York Times, December 1996] In a Boolean algebra, x + y + x + y = y implies x + y + x + y = y (McCune, 1996) • . . . . . . and ELAN

DTP2000 5

Two main kinds of rules

Computation rules

2 + 3 fib(33) sort(data-base) Compute as fast as possible the unique result

Deduction rules

Sequent calculus Resolution method Solve(2x + 4y − 3z − u = 0 in N) The deduction has to be guided

DTP2000 6

Deduction rule application needs to be controlled

Lazy evaluation Innermost/Outermost reduction Search plans Action plans Tactics User interaction ▶ This requires to search for a particular derivation corresponding to the desired control.

DTP2000 7

rewrite rewrite rewrite rewrite rewrite rewrite rewrite rewrite

Logic Programming, Theorem Proving, Constraint Solving, Expert Systems are instances of the same deduction schema:

Apply rewrite rules on formulas with some strategy, until getting specific forms

▶ Rewrite blindly: implements computations ▶ Rewrite wisely: implements deduction Poincaré principle [BarendregtBarendsen97] Deduction modulo [DowekHardinKirchner98]

DTP2000 8

How this presentation will proceed:

— Give a first introduction of ELAN with an emphasis on control of rewriting — Present the rewriting calculus as a semantical foundation for controled rule based systems — Come back on ELAN and its main features

DTP2000 9

Short presentation of ELAN

ELAN= computation rules + (deduction rules + strategies)

DTP2000 10

ELAN’s General features

• ELAN has been designed to prototype, experiment and study deduction systems, in particular for constraint solving and theorem proving.

• ELAN is a declarative rewriting language.

• ELAN has two main originalities w.r.t. other algebraic languages:

– non-deterministic computations – a user defined strategy language to control rewriting

DTP2000 11

Deductions and computations

• Rules for computations:

▶ unique normal form required ▶ leftmost innermost deterministic strategy fixed

• Rules for deductions:

▶ no confluence nor termination required ▶ application strategy required.

– non-determinism handled in practice by backtracking – strategies are used to express choices – strategy may fail

DTP2000 12

Example 1: Very simple ...

module fib_builtin
import global builtinInt;
  end

operators global
        fib(@) : (builtinInt) builtinInt ;
  end

rules for builtinInt
        n : builtinInt ;
global
    [] fib(0) => 1 end
    [] fib(1) => 1 end
    [] fib(n) => fib(n - 1) + fib(n - 2) if greater_builtinInt(n,1) end
  end
end

fib(33) = 5702887 11405773 rewrite steps in 0.695 s 16.411.184 rewrite/s

Digital 500/500, 128Mo

DTP2000 13

Example 2: Polynome derivative in ELAN

module poly1 import global int; sort poly;
operators global
  X : poly;
  @ : ( int ) poly;
  @ + @ : ( poly poly ) poly assocRight;
  deriv(@) : ( poly ) poly;
end
rules for poly
  p1, p2 : poly; n : int;
global
  [] deriv(X) => 1 end
  [] deriv(n) => 0 end
  [] deriv(p1+p2) => deriv(p1)+deriv(p2) end
end

DTP2000 14

module poly3[Vars]
import global int Vars identifier list[identifier];
sort variable poly;
operators global
  FOR EACH Id:identifier SUCH THAT
    Id:=(listExtract) elem(Vars) : { Id : variable; }
  @ : ( variable ) poly;
  @ : ( int ) poly;
  @ + @ : ( poly poly ) poly assocRight pri 1 (AC);
  (@ + @) : ( poly poly ) poly alias @ + @:;
  @ * @ : ( poly poly ) poly assocRight pri 2 (AC);
  (@ * @) : ( poly poly ) poly alias @ * @:;
  deriv(@,@) : ( poly variable ) poly;
end

... ... ...

DTP2000 15

A simple strategy

[] P+0 => P end
[] P*0 => 0 end
[] deriv(x,x) => 1 end
[] deriv(n,x) => 0 end
[] deriv(p1+p2,x) => deriv(p1,x) + deriv(p2,x) end

[factorize] P + n1*p1 + n2*p1 => P + n3*p1
                                      where n3:=()n1+n2 end
[factorize] P + (p1*p2) + (p1*p3) => P + p1*(p2+p3) end
[expand] P + (p1+p2)*p3 => P + p2*p3 + p1*p3 end
end
strategies for poly
  [] simplify =>
    normalize( one(expand) ) ; normalize( one(factorize) )
  end

DTP2000 16

Elementary strategies

Built from a few built-in constructors: - Sequential composition S1; S2 - Iteration

repeat*(S) or iterate*(S) - Choice points

dk(S1, . . . , Sn) : all results of all Si dc(S1, . . . , Sn) : all results of one Si first(S1, . . . , Sn) : all results of the first Si - Cut points

dc one(S1, . . . , Sn) : one result of one Si first one(S1, . . . , Sn) : one result of the first Si - Failure fail and Identity id

DTP2000 17

Another example: propositional sequent calculus H, P ⊢ Q H ⊢ ¬P, Q neg-r

rules for Seq
        P, Q, R : Pred; H : Pred; S1, S2 : Seq;
global
  [neg-r] H |- ^P : Q => S1
                                   where S1 := (dedstrat) H : P |- Q
  end

...

  [axio] P : H |- P : R => $
                               if neq_Pred(P, EmptyP)

DTP2000 18

The true code

Built (for later use) the proof term:

[negr] H |- ^P : Q => [#negd,H |- ^P : Q] <> S1
                               where S1 := (dedstrat) H : P |- Q
end

DTP2000 19

Strategies

strategies for Seq
implicit

        [] SetRules => first one(
                                axio
                                ,negd ,disjd
                                ,impd ,negg ,conjg
                                ,disjg ,conjd ,impg)
        end
end

strategies for Seq
implicit
        [] dedstrat => first one( Start );
                         repeat*( SetRules )
        end
end

DTP2000 20

The resulting proof term

[dedstrat](A |=> B |- ^(B) |=> ^(A))

evaluates to:

#infer[#impd]<(A#to B)#vdash(#neg(B)#to#neg(A))>
<#infer[#negd]<(A#to B),#neg(B)#vdash#neg(A)>
<#infer[#negg]<A,(A#to B),#neg(B)#vdash EmptyP>
<#infer[#impg]<A,(A#to B)#vdash B>
<#infer[#axiom]<A,B#vdash B><#mbox<>>&
#infer[#axiom]<A#vdash A,B><#mbox<>>>>>>
end

DTP2000 21

DTP2000 22

To summarize: Rules in ELAN

Rules are possibly labeled conditional rewrite rules with local variable affectations

[lab] : l(x) ⇒ r(y) if v where y := [S]u(x)

— lab is the rule label, — l and r the respective left and right-hand sides, — v the condition and — y := [S]u a local affectation, assigning to the local variable y one of the results of the strategy S applied to the term u.

DTP2000 23

Computation and Deduction in ELAN

Computations are described with unlabeled rules Deductions are described by labeled rules Strategies for deductions are themselves described using labeled or unlabeled rewrite rules.

DTP2000 24

The Computational System Tower ... ... Meta Meta Strategies Meta Strategies Strategies Terms

DTP2000 25

Rewriting with strategies IS now a realistic programming paradigm

On typical applications, ELAN applies per second:

• 10 millions of non-labeled rules • 1 million of labeled rules • 100 000 of AC rules On real size applications:

• Jobs shop 10x10: 2 billions rewrites applied in 2 hours

DTP2000 26

Some Applications developed in ELAN

• Theorem provers (KBs, Spike, B) • Automata computations • CP on finite domains • Unifications • Combination of deduction systems • . . .

DTP2000 27

N-queens queens(n, size) → x.ql if n > 0 where ql := (queens strat)queens(n − 1, size) where x := (iterate ∗ (dc(range rule)))size if noattack(1, x, ql) . . . Interpreter Compiler Compiler Query queens(8) queens(8) queens(11) Applied rules 128,949 128,949 19,286,638 Speed (inf/sec) 6,570 2,387,000 2,743,000 ▶ Note: local affectations and non-deterministic strategies ▶ Note: efficient management of non-deterministic strategies ▶ Speedup: 360

DTP2000 28

Completion à la Knuth Bendix

• 225 nonamed rules • 105 named rules • 43 strategies (8 dont know, 58 dont care, 6 iterate, 14 repeat) Interpreter Compiler Compiler Query group group p8 Applied rules 181,752 181,752 21,607,279 Speed (inf/sec) 1,452 826,145 919,458 ▶ Note: conditional rules, local affectations and non-deterministic strategies,. . . ▶ Speedup: 135

DTP2000 29

Nat10

Arithmetic rules introduced by Marché et all

• 56 rules rooted by the AC symbol + • 11 rules rooted by the AC symbol * • 82 syntactic rules Interpreter Compiler Compiler Query Fib(16) Fib(16) Fib(25) Applied rules 15,384 15,384 1,171,043 Speed (inf/sec) 44 49,625 53,717

DTP2000 30

How does it work?

• leftmost-innermost normalization • many-to-one matching • re-using left-hand-side parts • shared terms allowed • particular choice point management → What about the semantics?

DTP2000 31

The rewriting calculus as a semantics of ELAN

DTP2000 32

Towards a new calculus

That gives a first class status to:

▶ rewrite rules and ▶ strategies

DTP2000 33

The main ideas (1)

apply a rule at the top level of a term [l → r](t)

DTP2000 34

The main ideas (2)

The application operator [ ]( ) returns results that should be handled explicitely.

For example, if + is commutative, what is the result of the application of the rule x + y → x on a + b should it be a? or b? or a ⋄ b?

DTP2000 35

The main ideas (3)

— rule application — results object should be explicit objects of the calculus

DTP2000 36

The calculus ingredients

Five components:

1 The syntax of terms and substitutions, 2 The description of the substitution application on terms, 3 The matching algorithm used to bind variables to their actual arguments, 4 The evaluation rules describing the way the calculus operates locally. 5 The strategy describing how the evaluation rules operate globally.

DTP2000 37

1 ρ-Calculus Syntax

• elements in X (variables) and in T (F, X ) are ρ-terms, If t, u, t1, . . . , tm are ρ-terms and f ∈ Fm then the following are ρ-terms: • f(t1, . . . , tm) • {t1, . . . , tm} (if m = 0 we have the ρ-term ∅), • [t](u) (application of the ρ-term t to the ρ-term u), • t → u (rewrite rule). t ::= x | {t, . . . , t} | f(t, . . . , t) | [t](t) | t → t (x ∈ X , f ∈ F)

DTP2000 38

Examples of ρ-terms

• f(x, y) a first order term • f(x, x) → x a “standard” rewrite rule • [a → b](a) the application of the rule a → b to the term a • [f(x, y) → g(x, y)](f(a, b)) a classical rule application • [x → x](a) the result is {a} • [x → y](a) the result is {y} • [x → (x → x)]([x → y](a)) similar to the λ-term ((λx.λx.x) ((λx.y) a)) • [x → [x](x)](x → [x](x)) the well-known (Δ Δ) λ-term • [(a → b) → c](a) a more complicated ρ-term • [{a → b, a → c}](a) apply several rules

DTP2000 39

Handling results

Handling results via the set datastructure:

• has consequences on the calculus evaluation rules and their properties • could be done in a different ways using e.g. list, multisets, . . .

DTP2000 40

2 Substitution application on ρ-terms

Takes care of variable capture.

Grafting Substitution x → u x/u {x → a + y}(y → y + x) = y → y + (a + y) {x/a + y}(y → y + x) = y^′ → y^′ + (a + y)

DTP2000 41

4 ρ-Calculus: Evaluation Rules

The main rule in general: Fire [l → r](t) →→ r⟨⟨Solution(l ≪^?_T t)⟩⟩ For example Fire [f(x) → x](f(a)) →→ x⟨⟨Solution(f(x) ≪^?_T f(a))⟩⟩

In case of syntactic matching: Fire [l → r](σ(l)) →→ {σ(r)} Fire [l → r](t) →→ ∅ if l does not match t

DTP2000 42

Congruence rules on operators

Congruence [f(u1, . . . , un)](f(v1, . . . , vn)) →→ {f([u1](v1), . . . , [un](vn))} Congruence fail [f(u1, . . . , un)](g(v1, . . . , vm)) →→ ∅

Equivalent to the reduction of [f(x1, . . . , xn) → f([u1](x1), . . . , [un](xn))](t)

DTP2000 43

Congruence rules on sets

Distrib [{u1, . . . , un}](v) →→ {[u1](v), . . . , [un](v)}

Batch [v]({u1, . . . , un}) →→ {[v](u1), . . . , [v](un)}

SwitchL {u1, . . . , un} → v →→ {u1 → v, . . . , un → v}

SwitchR u → {v1, . . . , vn} →→ {u → vn, . . . , u → vn}

OpOnSet f(v1, . . . , {u1, . . . , um}, . . . , vn) →→ {f(v1, . . . , u1, . . . , vn), . . . , f(v1, . . . , um, . . . , vm)}

DTP2000 44

Handling sets of sets

Flat {u1, . . . , {v1, . . . , vn}, . . . , um} →→ {u1, . . . , v1, . . . , vn, . . . , um}

DTP2000 45

Applying substitutions

Meta-rules in this version Propagate r⟨⟨{s1, . . . , sn}⟩⟩ ⊏ {s1(r), . . . , sn(r)} Clash r⟨⟨∅⟩⟩ ⊏ ∅ Could be made explicit: ρσ-calculus

DTP2000 46

Remarks

ρ-calculus describes rewrite rules application at the top level of a term [f(x) → g(x)](f(b)) →→_{F ire}{g(b)} but [x + 0 → x](f(3 + 0)) →→_{F ire} ∅ and [f(x + 0 → x)](f(3 + 0)) →→_{Congruence} f([x + 0 → x](3 + 0)) →→_{F ire} f({3})

The evaluation rules expressing the behavior of ρ-calculus are applied *everywhere* in the term (e.g. like β-reduction).

DTP2000 47

3 Matching

In general For a theory T a T -match-equation is a formula of the form t ≪^?_T t^′, where t and t^′ are ρ-terms.

A substitution σ is a T -solution of t ≪^?_T t^′ when σ(t) =T t^′ T -matching is in general undecidable. Decidable cases:

• higher-order pattern matching [Miller-89] • higher-order matching (up to the fourth order [PadovaniThese-96]) • many first-order equational theories and their combinations

DTP2000 48

Solutions set

We define the function Solution on a T -matching system S as returning the set of: — all T -matches of S when S is not trivial — {Id} (the identity substitution) when S is trivial — ∅ when the matching algorithm fails.

DTP2000 49

Specificities of the ρ_T -calculus

• calculus of explicit rule application • get the “matching power” • explicit handling of result sets, • one of the main parameter of the calculus is the matching algorithm,

• allows the full control of the rewrite rule application • allows a uniform combination of higher-order and first-order features, • allows expressing strategies (e.g. search strategies)

• the ρ-calculus is quite different from the rewrite relation generated by a trs.

DTP2000 50

The ρ_∅-calculus

An instance of ρT -calculus where: — T = ∅ — the left members of matching equations are composed only of first-order terms (i.e. not containing arrows or applications).

From now on we only consider ρ_∅-calculus.

No difficulty to extend to the case of first-order equational matching (e.g. AC-matching)

DTP2000 51

Technical specificities

• A rule application always fires ... but the result may be the empty set (when the rule does not match an under-evaluated term), • Thus the wild ρ_∅-calculus is trivialy non-confluent: [x → b](∅) F ire � � � Batch � � � � {b} ∅

DTP2000 52

Undesired matching failures [a → b]([a → a](a)) F ire � � � �_{F ire} � � � � � [a → b]({a}) Batch � F ire � � � � � ∅ {[a → b](a)} F ire � ∅

{{b}} F lat � {b}

DTP2000 53

Non-right-linear rewrite rules [x → f(x, x)]({a, b}) F ire � Batch � � � � {f({a, b}, {a, b})} OpOnSet � {[x → f(x, x)](a), [x → f(x, x)](b)} F ire � {{f(a, {a, b}), f(b, {a, b})}} OpOnSet � {{f(a, a)}, {f(b, b)}} F lat � {{{f(a, a), f(a, b)}, {f(b, a), f(b, b)}}} F lat � {f(a, a), f(b, b)}

{f(a, a), f(a, b), f(b, a), f(b, b)}

DTP2000 54

ConfStratStrict

The term [l → r](t) is reduced using the evaluation rule Fire only if t ∈ T (F).

ConfStrat

The term [l → r](t) is reduced using the evaluation rule Fire only if:

• the term l is linear, • the term l weakly subsumes the term t (FPos(l) ⊆ FPos(t)), • the term t contains no empty set and no set with more than one element, • the term t contains no sub-term of the form [u](v) where u is not a rewriting rule, • for all sub-term [u → w](v) of t, u subsumes v, or • t ∈ T (F) is a closed first order term.

DTP2000 55

ConfStratOper

The term [l → r](t) is reduced using the evaluation rule Fire only if t ∈ T (F) or:

• the term l is linear and • the term l weakly subsumes the term t and – the rewrite rule l → r is conservative (F V (l) ⊆ F V (r)) or – the term t contains no empty set – the term t contains no sub-term of the form [u](v) where u is not a rewriting rule, – for all sub-term [u → w](v) of t, u subsumes v, and – the rewrite rule l → r is right linear or – the term t contains no set with more than one element.

DTP2000 56

When using one of the strategies ConfStratStrict, ConfStrat or ConfStratOper, the ρ_∅-calculus in confluent.

DTP2000 57

About expressiveness of the ρ-calculus

DTP2000 58

Encoding the λ-calculus t ::= x | {t} | f(t, . . . , t) | [t](t) | x → t

FirePropagate [x → r](t) →→ {{x/t}r}

φ(λx.t) = x → t φ((s t)) = [s](t) Given t and t^′ two λ-terms. Then, t −→β t^′ iff φ(t) −→ρ {φ(t^′)}. Proof: For any λ-term (λx.t u) that reduces to {x/u}t the corresponding ρ-reduction is [x → t](u)^∗ −→F ireP ropagate {{x/u}t}. □

DTP2000 59

Encoding rewriting t ::= x | {t} | f(t, . . . , t) | [t](l) | l → l

Given t and t^′ two terms in T (F, X ) and R a first order term rewrite system.

If t^∗−→R t^′ then there exists a ρ-term u constructed using the rules in R such that [u](t)^∗−→ρ {t^′}. Proof: Let us consider a rewrite rule (l → r) that applies on the term t at position p and thus, t[θl]p −→R t[θr]p = t^′. The ρ-term u to be applied in the ρ-calculus is t[l → r]p and we get: [t[l → r]p](t[θl]p)^∗−→ρ {t[θr]p} □

DTP2000 60

(Normalization) Strategies

Given a rewrite theory R does it exist a ρ-term ξR such that for any term u, if u reduces to the term v in the rewrite theory R then [ξR](u) ρ-reduces to {. . . , v, . . .}?

Given a rewrite theory R does it exist a ρ-term ξR such that for any term u, if u normalizes to the term v in the rewrite theory R then [ξR](u) ρ-normalizes to {. . . , v, . . .}?

DTP2000 61

Extending the calculus: with syntactic sugar t ::= id | fail | t; t | dk(t, t) Evaluation rules Identity id →→ x → x Fail fail →→ x → ∅ Compose [s1; s2](t) →→ [s2]([s1](t)) DK [dk(s1, s2)](t) →→ {[s1](t), [s2](t)}

DTP2000 62

The first operator

First [first(s1, . . . , sn)](t) →→ ⟨[s1](t), . . . , [sn](t)⟩

First fail ⟨∅, t1, . . . , tn⟩ →→ ⟨t1, . . . , tn⟩ First success ⟨t, t1, . . . , tn⟩ →→ {t} if t ≠ ∅ is closed and contains no redex First single ⟨⟩ →→ {}

DTP2000 63

Term traversal

Tseq [Φ(r)](f(u1, . . . , un)) →→ ⟨{f([r](u1), . . . , un)}, . . . , {f(u1, . . . , [r](un))}⟩ Tseq ct [Φ(r)](c) →→ ∅

Tpar [Ψ(r)](f(u1, . . . , un)) →→ {f([r](u1), . . . , [r](un))} Tpar ct [Ψ(r)](c) →→ {c}

DTP2000 64

Fix point operators

Θ = [A](A) with A = x → (y → [y]([[x](x)](y))) Thus [Θ](G)^∗−→ρ {[G]([Θ](G))}

Bottom-up application

Oncebu(r) ≡ [Θ](Hbu(r)) where Hbu(r) ≡ f → (x → [first(Φ(f), r)](x)) Ex: [Once_{bu}(a → b)](f(a, g(a)))=⇒ρ{f(b, g(a))}

DTP2000 65

Repeat operator

repeat∗ (r) ≡ [Θ](J(r)) where J(r) ≡ f → (x → [first(r; f, id)](x))

Normalization strategies

im(r) ≡ repeat∗ (Oncebu(r))

om(r) ≡ repeat∗ (Once_{td}(r)).

DTP2000 66

Encoding Conditional Rewriting

The conditional rewrite rule: l → r if c is simply encoded by l → [{T → r, F → ∅}]([lmim](c)) Or even simpler l → [T → r]([lmim](c))

DTP2000 67

ELAN’s semantics

Using rewriting logic [Meseguer 92] Using ρ-calculus

DTP2000 68

The ELAN basic object: rewrite rules with where

Rules are labelled conditional rewrite rules with local variable affectations [lab] l ⇒ r(x, y) where x := [S1]u1 if c1(x) where y := [S2]u2 if c2(x, y) — lab is the rule label, — l and r are the respective left and right-hand sides, — c1, c2 are the conditions and — z := [S]u are local affectations, assigning to the local variable z one of the result of the strategy S applied to the term u.

DTP2000 69

Example of ELAN’s rule

[deriveSum] p1 + p2 ⇒ p^′_1 + p^′_2 where p^′_1 := (derive)p1 where p^′_2 := (derive)p2 can be represented by the following two ρ-terms

p1 + p2 → [derive](p1) + [derive](p2)

p1 + p2 → [p^′_1 → [p^′_2 → p^′_1 + p^′_2]([derive](p2))]([derive](p1))

DTP2000 70

An expression of ELAN’s rules in ρ-calculus

[lab] l ⇒ r(x) where x := [S1]u1 if c1(x) The rule is expressed as the ρ-term:

l → [x → [T rue → r(x)]([lmim](c1(x)))]([S1](u1))

DTP2000 71

With two where if

[lab] l ⇒ r(x) where x := [S1]u1 if c1(x) where y := [S2]u2 if c2(x, y) The rule is expressed as the ρ-term:

l → [x → [ T rue → [ y → [T rue → r(x, y)]([lmim](c2(x, y))) ]([S2](u2)) ]([lmim](c1(x))) ]([S1](u1))

DTP2000 72

To sum-up

▶ ρT -calculus is a simple, elegant and powerful new calculus — explicit rule application (and thus “matching power”) — explicit handling of result sets — paramaterized by a matching theory T ▶ Clear distinction between the rewrite relation and the rewrite calculus. ▶ It allows uniform combination of first and higher-order computations ▶ ρ-calculus gives a simple semantics to ELAN programs

DTP2000 73

Back on

http://www.loria.fr/ELAN

Current release: V3.4 The ELAN team:

Peter Borovanský, Horatiu Cirstea, Hubert Dubois, Claude Kirchner, Hélène Kirchner, Pierre-Etienne Moreau, Christophe Ringeissen, Marian Vittek.

DTP2000 74

Back on Fibonacci

fib(0) → 1 fib(1) → 1 fib(n) → fib(n − 1) + fib(n − 2) fib(3) → fib(2) + fib(1) fib(2) + fib(1) → fib(2) + 1 fib(2) + 1 → fib(1) + fib(0) + 1 fib(1) + fib(0) + 1 → 1 + fib(0) + 1 . . .

DTP2000 75

In ELAN

module fib

import global int;
end

operators global
 fib(@) : (int) int ;
end

rules for int
n : int ;
global
    [] fib(0) => 1 end
    [] fib(1) => 1 end
    [] fib(n) => fib(n - 1) + fib(n - 2) if n > 1 end
end
end

DTP2000 76

We need to specify how to start a computation

LPL fib description
  query of sort int
  result of sort int
  import fib
  start with ()query
end

DTP2000 77

Then we can run it

-hilbert 63-> elan fib

******** ELAN version 3.3 (08/12/98.16:18) ********

 (c) LORIA (CNRS, INPL, INRIA, UHP, U-Nancy 2), 1994 - 1998
   handling module fib
     handling module int
       handling module builtinInt
         handling module bool
         end of bool
       end of builtinInt
     end of int
   end of fib

enter query term finished by the key word ’end’:
fib(4) end

[] start with term :

DTP2000 78

    fib(4)
[] result term:
    5
[] end
enter query term finished by the key word ’end’:

DTP2000 79

ELAN: Logical Organization

DTP2000 80

ELAN: Environment

DTP2000 81

The language: Main principles

• Rules may be non-terminating and non-confluent but then their application has to be controlled by strategies • Rules and strategies are first-class objects • Application of rule or a strategy on a term may give 0, 1, or several results (could be infinite)

DTP2000 82

The language

Close to algebraic specification formalism with additional specificities:

• Deductions and computations:

- Rules for computations (confluent and terminating) - Rules for deductions (no confluence nor termination required)

• A strategy language to express control and derivation tree exploration.

Non-determinism (sets of results) handled by backtracking.

• Generic language constructions handled by a pre-processor.

DTP2000 83

Programming styles

• Functional programs are naturally expressed • Non-determistic computations are simply described • Strategy programming for expressing the control

DTP2000 84

Modules

Functions, rules and strategies are defined in modules: .eln file Importations of already defined modules that may be local or global Predefined modules: bool,int,ident,list[X]...

module enum0
import global int list[int] ; end

DTP2000 85

Signatures

• Sorts • Functions with their ranks

sort int list[int];
end

operators global
  enum(@,@): (int int) list[int] ;
end

DTP2000 86

Rules

Conditional rewrite rules

rules for list[int]
    i,j,k,l: int ;
global
[] enum(i,j) => i.enum(i+1,j) if i<=j end
[] enum(i,j) => nil if i>j end
end

DTP2000 87

Top level

A .lgi file specifies the top module and the sorts of the query and result.

LPL enum0 description
  query of sort list[int]
  result of sort list[int]
  import enum0
  start with () query
end

DTP2000 88

******** ELAN version 3.3 (08/12/98.16:18) ********
 (c) LORIA (CNRS, INPL, INRIA, UHP, U-Nancy 2), 1994 - 1998
   handling module enum0
     handling module int
       handling module builtinInt
         handling module bool
         end of bool
       end of builtinInt
     end of int
     handling module list[int]
     end of list[int]
   end of enum0
enter query term finished by the key word ’end’: enum(3,6) end
[] start with term :
    enum(3,6)
[] result term:
    3.4.5.6.nil
[] end

DTP2000 89

Introducing strategies

Instead of having enum(3,6) as a list 3.4.5.6.nil

is it possible to generate successively

3, 4, 5, 6?

DTP2000 90

Labelled rules

Rules are given a label.

rules for int
  i,j,k,l: int ;
global
[final] enum(i,j) => i end
[iter] enum(i,j) => enum(i+1,j)
                  if i<j end
end

DTP2000 91

Strategy definition

A strategy operator defined by a rewrite rule on a strategy sort.

stratop global
  enumStrat1 : <int->int> bs ;
end

strategies for int
implicit
[] enumStrat1 => iterate*(iter) ; final end
end

DTP2000 92

******** ELAN version 3.3 (08/12/98.16:18) ********
 (c) LORIA (CNRS, INPL, INRIA, UHP, U-Nancy 2), 1994 - 1998
   handling module enum1
   ...
enter query term finished by the key word ’end’:
enum(3,6) end
[] start with term :
    enum(3,6)
[] result term:
    3
[] result term:
    4
[] result term:
    5
[] result term:
    6
[] end

DTP2000 93

How is it computed?

[final] enum(i,j) => i
[iter] enum(i,j) => enum(i+1,j)
                  if i<j

[] enumStrat1 => iterate*(iter) ; final

                                           (enumStrat1) enum(3,6)

iterate iter final iterate0 enum(3, 6) 3 iterate1 enum(4, 6) 4 iterate2 enum(5, 6) 5 iterate3 enum(6, 6) 6 iterate4 fail fail

DTP2000 94

Elementary strategies

Built from a few built-in constructors: - Labelled rule [ℓ] - Sequential composition S1; S2 - Iteration repeat*(S) or iterate*(S) repeat*(S) applies repeatedly the strategy S until it fails and returns the results of the last unfailing application.

iterate*(S) is similar to repeat*(S) but returns all intermediate results of repeated applications.

DTP2000 95

- Choice points dk(S1, . . . , Sn) : all results of all Si dc(S1, . . . , Sn) : all results of one Si first(S1, . . . , Sn) : all results of the first Si - Cut points dc one(S1, . . . , Sn) : one result of one Si first one(S1, . . . , Sn) : one result of the first Si - Failure fail and Identity id

DTP2000 96

Labelled rules

module tree
import global int ; end
sort int tree ; end
operators global
        @ : (int) tree;
        leaf(@) : (int) tree;
        node(@,@,@) : (tree int tree) tree;
end
rules for tree
        l, r: tree; n: int;
global
[Left] node(l,n,r) => l end
[Right] node(l,n,r) => r end
[L_value] leaf(n) => n end
end

DTP2000 97

Strategies

stratop
  global
        allleaves : <tree -> tree> bs;
end

strategies for tree
implicit
  [] allleaves => iterate*(dk(Left,Right));L_value end
end
end

DTP2000 98

******** ELAN version 3.3 (08/12/98.16:18) ********
 (c) LORIA (CNRS, INPL, INRIA, UHP, U-Nancy 2), 1994 - 1998
   handling module tree
   ...
enter query term finished by the key word ’end’:
node(node(leaf(1),2,leaf(3)),4,node(leaf(5),6,leaf(7))) end
[] start with term :
    node(node(leaf(1),2,leaf(3)),4,node(leaf(5),6,leaf(7)))
[] result term:
    1
[] result term:
    3
[] result term:
    5
[] result term:
    7
[] end

DTP2000 99

Rules with local variables and patterns

Labelled rules are always applied at top position. Local variables used to direct strategy application

rules for poly
  p1, p2, p3, p4, p5 : poly; x variable;
global
  [] deriv(p1*p2,x) => p5
        where p3:=(simplify) deriv(p1,x)*p2
        where p4:=(simplify) p1*deriv(p2,x)
        where p5:=(simplify) p3+p4 end
end
strategies for poly
  [] simplify => repeat*(first one(expand)) ;
                 repeat*(first one(factorize))
end

DTP2000 100

Rules with local patterns

Local variables generalise to patterns (a term with variables)

[] move(S) => C(x,y)
      where < x , y > := position(S)
      if x=y

DTP2000 101

General form of rules

[ℓ] l → r where p1 := (S1)c1 . . . where pn := (Sn)cn • l, r, p1, . . . , pn, c1, . . . , cn ∈ T (F, X ), • Var(()pi) ∩ (Var(()l) ∪ Var(()p1) ∪ · · · ∪ Var(()pi−1)) = ∅, • Var(()r) ⊆ Var(()l) ∪ Var(()p1) ∪ · · · ∪ Var(()pn) and • Var(()ci) ⊆ Var(()l) ∪ Var(()p1) ∪ · · · ∪ Var(()pi−1). where true := c is written if c. pi often reduced to a variable x.

Si may be the identity strategy, written ().

DTP2000 102

Application of a rule

[ℓ] l → r where p1 := (S1)c1 . . . where pn := (Sn)cn

To apply the rule to a subject t, the matching substitution from l to t (lσ = t) is successively composed with each matching μi from pi to (Si)ciσμ1 . . . μi−1, for i = 1, . . . , n. Evaluation of (S)t:

1. t is normalized using the unlabelled rules 2. One tries to apply a labelled rule according to the strategy S. Choice points are set when there are several results. Backtrack occurs when the set of results in empty.

DTP2000 103

Rules with Associative Commutative functions

If l or p contains AC function symbols, AC matching is called and can return several solutions.

One may want to get all solutions of an AC matching problem and build all possible results of rewriting with these different matching substitutions. When the rule contains a sequence of matching conditions, failing to satisfy the i-th condition causes a backtracking to the previous one.

DTP2000 104

operators global
  @ U @ :(set set) set (AC) ;
  emptySet : set ;
  (@): (int) set ;
  extract(@) : (set) int ;
end
rules for int
   i: int ; s: set ;
global
[Rule] extract( (i) U s ) => i end
end

Application of the rule [Rule] to

extract(emptySet U (1) U (2) U (3) U (4) U (5))

returns one of 1,2,3,4,5.

Application of the strategy dk(Rule) returns all these results successively.

DTP2000 105

module enumAC3
import global int ; end
sort set ; end

operators global
  @ U @ :(set set) set (AC) ;
  emptySet : set ;
  (@): (int) set ;
  enum(@,@): (int int) set ;
  extract(@) : (set) int ;
end

stratop global
  enumStratAC : <int -> int> bs ;
end

DTP2000 106

rules for set
  i,j: int ;
global
[] enum(i,j) => emptySet if i>j end
[] enum(i,j) => (i) U enum(i+1,j)
                       if i <= j end
end
rules for int
  i: int ; s: set ;
global
[Rule] extract( (i) U s ) => i end
end
strategies for int
implicit
[] enumStratAC => dk(Rule) end
end

DTP2000 107

******** ELAN version 3.3 (08/12/98.16:18) ********
 (c) LORIA (CNRS, INPL, INRIA, UHP, U-Nancy 2), 1994 - 1998
   handling module enumAC3
   ...
enter query term finished by the key word ’end’:
enum(3,6) end
[] start with term :
    extract(enum(3,6))
[] result term:
    6
[] result term:
    5
[] result term:
    4
[] result term:
    3
[] end

DTP2000 108

N queens • • • • is represented by the list (3,1,0,2)

DTP2000 109

Module nqueensAC

operators
 global
    queensAC(@) : (int) list[int] ;
 local
    queens(@,@) : (set list[int]) list[int] ;
    ok(@,@,@) : ( int int list[int] ) bool;
    @ U @ : (set set) set (AC);
    (@) : (int) set ;
    [@ U @] : (int set) set ;
    Set(@) : (int) set ;
    Empty : set ;
end

DTP2000 110

stratop
  global
    queens : <list[int] -> list[int]> bs;
  local
    extractPos : <set -> set> bs;
end

DTP2000 111

rules for set
  S: set ;
  i: int ;
global
  [] Set(0) => Empty U (0) end
  [] Set(i) => Set(i-1) U (i) end
  [extractrule] (i) U S => [i U S] end
end

strategies for set
implicit
  [] extractPos => dk(extractrule) end
end

DTP2000 112

rules for bool
  p,d,diff : int;
  l : list[int];
global
  [] ok(diff,d,nil) => true end
  [] ok(diff,d,p.l) => false if d-p == diff end
  [] ok(diff,d,p.l) => false if p-d == diff end
  [] ok(diff,d,p.l) => ok(diff+1,d,l) end
end

DTP2000 113

rules for list[int]
  n:int;
global
  [] queensAC(n) => queens(Set(n-1),nil)

rules for list[int]
  p1: int; s,s1:set ; l,l1: list[int];
local
  [final] queens(Empty,l) => l end
  [queensrule] queens(s,l) => queens(s1,p1.l)
                  where (set) [p1 U s1] :=(extractPos) s
                  if ok(1,p1,l)

strategies for list[int]
implicit
  [] queens => repeat*(dk(queensrule)); dc(final)

DTP2000 114

Execution

Query term: queensAC(4) end
1.3.0.2.nil
2.0.3.1.nil
2 solutions

Query term: queensAC(8) end
...
3.1.6.2.5.7.4.0.end
92 solutions

DTP2000 115

Execution with the compiler

On a Pentium II 450mhz, linux, with -optimiseChoicePoint.

queens(8) -- 92 results
rewrite_step = 106049
0.150u 0.020s

queens(10) -- 724 results
rewrite_step = 2608909
1.850u 0.230s

queens(12) -- 14200 results
rewrite_step = 82711249
51.230u 2.090s

queens(14) --- 365596 results
rewrite_step = 3307192557
1892.910u 60.260s

DTP2000 116

Defined Strategies

Strategy operators with arguments (parameters).

map : (⟨s⟩)⟨list[s]⟩

map(S) ⇒ dc(nil, S · map(S)) where S : ⟨s⟩ and s ∈ S. equivalent to [map(S)]nil ⇒ nil [map(S)]x · l ⇒ [S]x · [map(S)]l

DTP2000 117

module map[X]

import global
  strat[X] strat[list[X]] X list[X]; end
stratop global
  map(@) : (<X->X>) <list[X]->list[X]>;
  mac(@) : (<X->X>) <list[X]->list[X]>;
end

rules for X
 x : X; n,m : int;
global
  [mul2] x => x*2 end
end

DTP2000 118

strategies for list[X]
 s : <X->X>; h : X; h1 : X; t : list[X]; t1 : list[X];
explicit
[.] [mac(s)]nil => nil end
[.] [mac(s)]h.t => h1.t1
       where h1:=[s]h
       where t1:=[mac(s)]t end
implicit
[.] map(s) => dc(nil,cons(s,map(s))) end
end
end

DTP2000 119

******** ELAN version 3.3 (08/12/98.16:18) ********
 (c) LORIA (CNRS, INPL, INRIA, UHP, U-Nancy 2), 1994 - 1998
   handling module int
   ....
   end of map[int]

enter query term finished by the key word ’end’:
    3.4.5.6.nil
[] start with term :
    3.4.5.6.nil
[] result term:
    6.8.10.12.nil
[] end

DTP2000 120

Strategy terms

• Elementary strategies: built from strategy constructors, applied with ( ) (built-in).

• Defined strategies: built from user’s strategy operators, applied with [ ] (meta-interpreter).

DTP2000 121

Strategy application

• Evaluation of (S)t:

1.t is normalized using the unlabelled rules 2.One tries to apply a labelled rule according to the strategy S.

• Evaluation of [S]t:

1.S and t are normalized using the unlabelled rules 2.rules defining [ ] are applied according to a strategy eval (defined in the meta-interpreter).

DTP2000 122

Rules on strategy terms

• When the rule is labelled by [.], it is used by the strategy meta-interpreter (ELAN program) according to its strategy eval.

• When the rule is labelled by [ℓ], it is used by the ELAN interpreter as any other labelled rule governed by a strategy.

• If the rule is unlabelled, the leftmost innermost predefined strategy is applied.

DTP2000 123

Deductions modulo Computations

– Labelled rules whose application is fully controled by the user elementary or defined strategies and,

– Unlabelled rules applied using a built-in leftmost innermost strategy (confluent and terminating).

Deductions, performed by labelled rules under strategies versus Computations, performed by unlabelled rules.

DTP2000 124

Some Applications developed in ELAN

• Theorem provers (KB completion, B provers) • Automata computations and approximations • COLETTE: Constraint Satisfaction Problems • Unification algorithms • CLP, Combination of solvers • Production rules (planification)

DTP2000 125

Why is this kind of programming language useful?

• Human thinking is very often of transitional kind • The underlined semantics is well defined and mathematicaly founded – uniform integration of first and higher-order features – rewriting calculus – rewriting logic • It allows us to set up safe programming environment ...

DTP2000 126

Example: Is this program safe?

module quotientRemainder

sort nat pairNat ;
end

operators global
 0 : nat ;
 s(@) : (nat) nat ;
 [@,@] : (nat nat) pairNat ;
 s(@) : (pairNat) pairNat ;
 q(@,@) : (nat nat) pairNat ;
 qq(@,@,@,@) : (nat nat nat nat) pairNat ;
// First argument of qq is number to be divided minus 1 (when called)
// Second argument is the divider minus 1 (when called)
// Third argument is the divider (just to remember)
// Last argument computes the potential remainder
end

DTP2000 127

rules for pairNat
x, y, z, r : nat ;
global
    [] q(0,s(y)) => [0,0] end
    [] q(s(x),s(y)) => qq(x,y,s(y),s(0)) end
    [] qq(x,0,z,r) => s(q(x,z)) end
    [] s([x,y]) => [s(x),y] end
    [] qq(0,s(y),z,r) => [0,r] end
    [] qq(s(x),s(y),z,r) => qq(x,y,z,s(r)) end
end

DTP2000 128

Proof assertion

Theorem QR in quotientRemainder
forall a: nat, b: nat, a = b * pr1(q(a,b)) + pr2(q(a,b))
// where pr1 and pr2 are the respective projections
and 0 =< pr2(q(a,b)) < b
end

Then not only one can apply the rules in order to compute the quotient and remainder of any two naturals, but on can also prove that

quotientReste |= ∀a : nat, ∀b : nat, a = b ∗ pr1(q(a, b)) + pr2(q(a, b))

∧ 0 ≤ pr2(q(a, b)) < b

DTP2000 129

Not yet available!

The full integration of the specification and proof levels is not yet available but practical examples are – already reachable and – quite appealing with respect to existing technologies For example the NSPK verification:

DTP2000 130

Encoding the Needham-Schroeder public-key protocol in ELAN

• user-definable number of initiators and responders • user-definable size of the network • the strategies guiding the rewrite rules describe a form of model-checking in which all the possible behaviors are explored

DTP2000 131

Data structures

• The agent

@ + @ + @ : ( AgentId SWC Nonce ) Agent;
@ : ( Agent ) listAgent;
@ || @ : ( listAgent listAgent ) listAgent (AC);

• The messages

@-->@:@[@,@,@] : ( AgentId AgentId Key Nonce Nonce Address ) message;
@ : ( message ) network;
@ & @ : ( network network ) network (AC);

• The intruders

@ # @ # @ : ( AgentId listNonce network ) intruder;

• The global state

@ <> @ <> @ <> @ : ( listAgent listAgent intruder network ) state;

DTP2000 132

Rewrite rules for the agents

• initiator starts the communication with a responder

[initiator-1]
   x+SLEEP+resp || E <> D <> w#l#ll <> ls =>
   x+WAIT+N(x,y) || E <> D <> w#l#ll <> x-->y:K(y)[N(x,y),DN,A(x)] & ls
                 where (Agent)y+std+init :=(extAgent) elemIA(D || w+SLEEP+DN)
end

• responder reads the message and sends the acknowledgement

[responder-1]
   E <> y+SLEEP+init || D <> I <> w-->y:K(y)[N(n1,n3),N(n2,n4),A(z)] & ls =>
   E <> y+WAIT+N(y,z) || D <> I <> y-->z:K(z)[N(n1,n3),N(y,z),A(y)] & ls
end

DTP2000 133

Rewrite rules for the agents

• initiator receives the acknowledgement and checks its validity

[initiator-2]
 x+WAIT+N(x,v) || E <> D <> I <> w-->x:K(x)[N(n1,n3),N(n2,n4),A(z)] & ls =>
 S
  choose
    try
      if x==n1 and v==n3
      where S:=() x+COMMIT+N(x,v) || E <> D <> I <> x-->v:K(v)[N(n2,n4),DN,DA] & ls
    try
      if x!=n1 or v!=n3
      where S:=() ERROR
  end
end

DTP2000 134

Rewrite rules for the intruder

• the intruder intercepts all the messages in the network but the messages generated by itself and stores or decrypts them.

[intruder-1]
   E <> D <> w#l#ll <> z-->x:K(w)[N(n1,n3),N(n2,n4),A(v)] & ls =>
   E <> D <> w#N(n1,n3) | N(n2,n4) | l#ll <> ls
                 if w!=z
end

• the nonces obtained previously by the intruder are used in order to generate fake messages that are sent to all the agents.

[intruder-4]
   E <> D <> w # resp | l # ll <> ls =>
   E <> D <> w # l # ll <> w-->y:K(y)[resp,DN,A(xadd)] & ls
                where (Agent)y+std+dn :=(extAgent) elemIA(D || E)
                where (Agent)xadd+std1+dn1 :=(extAgent) elemIA(D || E)
end

DTP2000 135

The invariants

• authenticity of the responder: if an initiator x committed with a responder y, then

y has really been involved in the protocol.

[attack-1] x+COMMIT+N(x,y) || E <> D <> I <> ls =>
           ATTACK
                if y!=i
                if not(existAgent(y+WAIT+N(y,x),D)) and
                   not(existAgent(y+COMMIT+N(y,x),D))
end

• authenticity of the initiator: if a responder y committed with an initiator x then the initiator have committed as well with y.

[attack-2] E <> y+COMMIT+N(y,x) || D <> I <> ls =>
           ATTACK
                if x!=i
                if not(existAgent(x+COMMIT+N(x,y),E))
end

DTP2000 136

The strategy

We apply repeatedly all the rewrite rules in any order and in all the possible ways until one of the attack rules can be applied.

[]attStrat => repeat*(
                   dk(
                      attack-1, attack-2,
                      intruder-1, intruder-2, intruder-3, intruder-4,
                      initiator-1, initiator-2, responder-1, responder-2)
                );
                attackFound
end

where

[attackFound] ATTACK => ATTACK end

DTP2000 137

The attack

I.1. A → I : {NA, A}_{K(I)} II.1. I(A) → B : {NA, A}_{K(B)} II.2. B → I(A) : {NA, NB }_{K(A)} I.2. I → A : {NA, NB }_{K(A)} I.3. A → I : {NB }_{K(I)} II.3. I(A) → B : {NB }_{K(B)}

DTP2000 138

A second strategy

[]attStrat => repeat*(
                        dk(
                           attack-1, attack-2,
                           initiator-1, initiator-2, responder-1, responder-2,
                           intruder-1, intruder-2, intruder-3, intruder-4)
                );
                attackFound
end

I.1. A → I : {N_A, A}_{K(I)} II.1. I(A) → B : {N_A, A}_{K(B)} I+II.2. B → A : {N_A, N_B }_{K(A)} I.3. A → I : {N_B}_{K(I)} II.3. I(A) → B : {N_B}_{K(B)}

DTP2000 139

The corrected protocol

1. A → B: {N_A, A}_{K(B)} 2. B → A: {N_A, N_B, B}_{K(A)} 3. A → B: {N_B }_{K(B)}

Modified rule: initiator-2

[initiator-2] x+WAIT+N(x,v) || E <> D <> I <> w-->x:K(x)[N(n1,n3),N(n2,n4),A(z)] & ls =>
S
  choose
    try
      if v==z // expected responder
      if x==n1 and v==n3
      where S:=() x+COMMIT+N(x,v) || E <> D <> I <> x-->v:K(v)[N(n2,n4),DN,DA] & ls
    try
      if v!=z // not expected responder
      if x!=n1 or v!=n3
      where S:=() ERROR
  end end

DTP2000 140

Optimizations of the current implementation

• each message is given a type according to its relevant fields (nonce and address), • initiator and responder listen only to messages coming from the intruder, • the intruder sends fake messages only to those agents that are able to handle it

DTP2000 141

Comparison with Murφ

ini res net Murφ ELAN simple ELAN(Murφ like) ELAN optimized 1 1 1 2578 rules 7442 rules 4308 rules 711 rules 0.56s 0.167s 0.125s 0.064s 1 1 2 136273 rules 453514 rules 333552 rules 6700 rules 18.40s 7.007s 4.980s 0.162s 2 1 1 25701 rules 575101 rules 257087 rules 26011 rules 6.81s 8.571s 3.946s 0.483s 2 2 1 557430 rules 118214389 rules 22985807 rules 753785 rules 303.46s 1658.296s 333.096s 12.392s

DTP2000 142

Protocol verification: some conclusions

• computational systems can be used as a logical framework for representing the Needham-Schroeder public-key protocol,

• this approach can be easily extended to other authentication protocols (TMN),

• the strategy used for guiding the application of the rewrite rules is important when an attack on the protocol exists,

• the behavior of intruders can be easily modified by changing the corresponding rules,

• the state space is finite and thus, some other methods should be used in order to show that properties proved for the finite model can be lifted to an unbounded model,

DTP2000 143

ELAN Version 3.4 (Mars 2000) http://www.loria.fr/ELAN

- An interpreter of the language, - A very efficient compiler (runs upto 15 million rules per second), - User’s manual and library documentation, - Examples; from tiny to very large ones.

ELAN is written in C++ and the current compiler is written in JAVA. The system runs on Suns and Digitals under Unix as well as on PCs under Linux.

We maintain elan-users-request@loria.fr, a low-volume moderated mailing list for announcements and discussion regarding ELAN.

DTP2000 144

Some ongoing works on ELAN and the ρ-calculus

• Cooperation CoQ – ELAN • Constraints as first class objects • ELAN and production rule systems • typed versions of the ρ-calculus • ρ-calculus with explicit substitutions

DTP2000 145

Next: adventures in Deduction Modulo